

Gliederung des Kapitels 1.1

1.1 Algorithmen und Sprachen

~~1.1.1 Darstellung von Algorithmen~~

~~1.1.2 Grundlegende Datenbereiche~~

~~1.1.3 Realisierte Abbildung~~

~~1.1.4 (Künstliche) Sprachen~~

~~1.1.5 Grammatiken~~

1.1.6 BNF, Syntaxdiagramme

1.1.7 Sprachen zur Beschreibung von Sprachen

1.1.8 Übungsaufgaben

07.11.02

Kap.1.1, Informatik I, WS 02/03

209

Abschnitt 1.1.6 soll Ihnen folgende Inhalte vermitteln:

Programmiersprachen lassen sich weitgehend durch kontextfreie Grammatiken beschreiben. Bereits 1958 wurde eine Variante, die BNF (sog. Backus-Naur-Form), zur Definition von Programmiersprachen eingesetzt; diese wurde später erweitert und durch Syntaxdiagramme grafisch veranschaulicht.

In diesem Abschnitt erlernen Sie die Erweiterung EBNF und wie man mit ihrer Hilfe die Syntax einer Programmiersprache definiert. Als Beispiele werden Teile von Ada 95 verwendet.

07.11.02

Kap.1.1, Informatik I, WS 02/03

210

1.1.6 BNF, Syntaxdiagramme

Definition 1.1.6.1: Backus-Naur-Form

Eine **BNF** ist ein Viertupel (V, Σ, P, S) mit

- (1) V ist eine nicht-leere endliche Menge (die Menge der **Nichtterminalzeichen**); alle Elemente sind von der Form $\langle \text{Zeichenkette} \rangle$ (in "Zeichenkette" treten '<' und '>' nicht auf),
- (2) Σ ist eine nicht-leere endliche Menge (die Menge der **Terminalzeichen**) mit $V \cap \Sigma = \emptyset$ und $|\Sigma| \geq 2$,
- (3) $S \in V$ ist ein Nichtterminalzeichen (das **Startsymbol**),
- (4) $P \subset V \{::=\} (V \cup \Sigma)^* (\{ \} (V \cup \Sigma)^*)^*$ ist eine endliche Menge (die Menge der **Regeln** oder **Produktionen**).

Eine BNF ist eine kontextfreie Grammatik (vgl. Def. 1.1.5.1), bei der die Nichtterminalzeichen durch Namen der Form " $\langle \text{Zeichenkette} \rangle$ " genauer bezeichnet werden können und bei der alle kontextfreien Regeln, die die gleiche linke Seite besitzen, zu einer Regel zusammengefasst werden, wobei die verschiedenen rechten Seiten durch einen senkrechten Strich getrennt werden.

Statt des Ersetzungspfeils " \rightarrow " schreibt man " $::=$ " (gesprochen: "Doppel-Doppel-Punkt-Gleich"; die Bedeutung ist: "kann ersetzt werden durch").

Bei Programmiersprachen verwendet man gerne das Nichtterminalzeichen $\langle \text{program} \rangle$ als Startsymbol, aus dem man die zulässigen Programme herleitet.

07.11.02

Kap.1.1, Informatik I, WS 02/03

211

07.11.02

Kap.1.1, Informatik I, WS 02/03

210

Beispiel 1.1.6.2: Definition der Bezeichner, vgl. Beispiel 1.1.5.5.

$\langle \text{Buchstabe} \rangle ::= A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P | Q | R | S | T | U | V | W | X | Y | Z | a | b | c | d | e | f | g | h | i | j | k | l | m | n | o | p | q | r | s | t | u | v | w | x | y | z$

$\langle \text{Ziffer} \rangle ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9$

$\langle \text{Zeichen für Bezeichner} \rangle ::= _ | \langle \text{Buchstabe} \rangle | \langle \text{Ziffer} \rangle$

$\langle \text{Bezeichner} \rangle ::= \langle \text{Buchstabe} \rangle |$

$\langle \text{Bezeichner} \rangle \langle \text{Zeichen für Bezeichner} \rangle$

$V = \{ \langle \text{Bezeichner} \rangle, \langle \text{Zeichen für Bezeichner} \rangle, \langle \text{Buchstabe} \rangle, \langle \text{Ziffer} \rangle \}$

$\Sigma = \{ A, B, C, D, E, F, G, H, I, J, K, L, M, N, O, P, Q, R, S, T, U, V, W, X, Y, Z, a, b, c, d, e, f, g, h, i, j, k, l, m, n, o, p, q, r, s, t, u, v, w, x, y, z, 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, _ \}$

Das Startsymbol ist hier $\langle \text{Bezeichner} \rangle$. ■

07.11.02

Kap.1.1, Informatik I, WS 02/03

213

Umwandlung einer kontextfreien Grammatik in die BNF:
Gegeben sei eine kontextfreie Grammatik (V', Σ, P', S') .

1. Ersetze jedes Nichtterminalzeichen $A' \in V'$ durch einen aussagekräftigen Namen A der Form $\langle \text{Zeichenkette} \rangle$, wobei die Zeichen '<' und '>' in dieser "Zeichenkette" nicht verwendet werden dürfen. Dies ergibt die neue Menge der Nichtterminalzeichen V mit dem Startsymbol S (anstelle S').
2. Ersetze in P' jedes Nichtterminalzeichen durch seinen neuen Namen der Form $\langle \text{Zeichenkette} \rangle$.
3. Wenn $A' \rightarrow u_1, A' \rightarrow u_2, \dots, A' \rightarrow u_k$ alle Regeln in P' mit A' als linker Seite sind und wenn A das zu A' gehörende neue Nichtterminalzeichen aus V ist, so ersetze die k Regeln durch $A ::= u_1 | u_2 | \dots | u_k$
So entsteht aus P' die neue Regelmengemenge P .

07.11.02

Kap.1.1, Informatik I, WS 02/03

214

Umgekehrt kann man mit dieser Anleitung auch eine BNF in eine "normale" kontextfreie Grammatik zurück verwandeln.

Der Ableitungsbegriff der BNF ist daher der gleiche wie der für kontextfreie Grammatiken. Somit sind die Ableitungsrelationen \Rightarrow und \Rightarrow^* sowie die erzeugte Sprache $L(G)$ auch für die BNF definiert (vgl. 1.1.5.2 und 1.1.5.3).

Die BNF wird nun um einige hilfreiche Abkürzungen zur EBNF erweitert.

07.11.02

Kap.1.1, Informatik I, WS 02/03

215

Verwenden von Schlüsselwörtern:

Schlüsselwörter der Sprache werden als Zeichenketten dargestellt, die in Apostrophe eingeschlossen werden. (Tritt im Schlüsselwort ein Apostroph auf, so wird dies durch zwei Apostrophe dargestellt.)

Beispiel:

<Boolesche Operatoren> ::= 'and' | 'or'

Hierdurch wird ausgedrückt, dass die Operatoren **and** und **or** eigentlich einzelne Terminalzeichen sind, jedoch als Zeichenkette durch Konkatenation anderer Zeichen dargestellt werden. Man könnte ansonsten in BNF auch schreiben:

<And-Operator> ::= and

<Or-Operator> ::= or

<Boolesche Operatoren> ::= <And-Operator> | <Or-Operator>

07.11.02

Kap.1.1, Informatik I, WS 02/03

216

Einführen von eckigen Klammern ("ein- oder keinmal"): Symbole oder Folgen von Symbolen, die auch wegfallen dürfen, werden in eckige Klammern eingeschlossen.

Beispiele:

<Ziffernfolge> ::= <Ziffer> [<Ziffernfolge>]

ist die Abkürzung für

<Ziffernfolge> ::= <Ziffer> | <Ziffer> <Ziffernfolge>

<Alternative> ::= 'if' <Boolescher Ausdruck>

'then' <Anweisung>

['else' <Anweisung>] 'fi'

07.11.02

Kap.1.1, Informatik I, WS 02/03

217

Einführen von geschweiften Klammern ("Iteration"):

Symbole oder Folgen von Symbolen, die beliebig oft (auch keinmal) wiederholt werden dürfen, werden in geschweifte Klammern eingeschlossen.

Beispiele:

<Ziffernfolge> ::= <Ziffer> { <Ziffer> }

ist die Abkürzung für

<Ziffernfolge> ::= ϵ | <Ziffer> <Ziffernfolge>

<Ziffernfolge> ::= <Ziffer> <Ziffernfolge>

Die Definition von Bezeichnern nach 1.1.6.2 lässt sich mit geschweiften Klammern wie folgt abkürzen:

<Bezeichner> ::= <Buchstabe> { _ | <Buchstabe> | <Ziffer> }

07.11.02

Kap.1.1, Informatik I, WS 02/03

218

Definition 1.1.6.3: Erweiterte Backus-Naur-Form

Die **EBNF** ist die Erweiterung der BNF um die Möglichkeiten:

Apostrophe für Schlüsselwörter als Terminalzeichen.

Eckige Klammern mit der Bedeutung: Die eingeschlossenen Teile können hier ein Mal auftreten, müssen aber nicht.

Geschweifte Klammern mit der Bedeutung: Die hierdurch eingeschlossenen Teile können beliebig oft nacheinander auftreten. (Dies entspricht dem Operator * auf Wortmengen.)

Ergänzender Hinweis: Falls eckige oder geschweifte Klammern als Terminalzeichen in den Regeln auftreten können, so bleibt es dem Ersteller der Regeln überlassen, für Eindeutigkeit zu sorgen.

07.11.02

Kap.1.1, Informatik I, WS 02/03

219

Entscheidendes Kriterium, ob eine Grammatik oder BNF für die Praxis geeignet ist, ist, dass mit ihr möglichst viele der Operationen für Sprachen nachgebildet werden können (vgl. 1.1.4.6). Für die EBNF gilt zum Beispiel:

Die **Vereinigung** wird durch '|' dargestellt, die **Konkatenation** durch Hintereinanderschreiben und die **Iteration** durch die geschweiften Klammern.

Die Operatoren "Durchschnitt" und "Komplement" erhalten wir auf diese Weise nicht. Für kontextfreie Sprachen gelingt dies auch nicht, da die Klasse der kontextfreien Sprachen nicht gegen Durchschnitt und nicht gegen Komplement abgeschlossen ist, d.h.: Wenn L_1 und L_2 kontextfreie Sprachen sind, dann sind der Durchschnitt $L_1 \cap L_2$ und das Komplement von L_1 im Allgemeinen keine kontextfreien Sprachen.

07.11.02

Kap.1.1, Informatik I, WS 02/03

220

Beispiel 1.1.6.4: Darstellung zur Basis b für $2 \leq b \leq 16$ in Ada.

Diese Zahlen werden in der Form

$2 \# 10010 \#$ oder $8 \# 60175 \#$ oder $16 \# A2F03 \#$

dargestellt, wobei zwischen den $\#$ -Zeichen mindestens eine Ziffer stehen muss und nur die Basen von 2 bis 16 zugelassen sind. Ansatz, um dies mit einer EBNF zu beschreiben:

$\langle \text{Darstellung mit Basis} \rangle ::= \langle \text{Basis} \rangle \# \langle \text{Ziffernfolge} \rangle \#$

Doch dieser Ansatz wird nicht erfolgreich sein, da zu jeder Basis eine andere Menge von Ziffern gehört. Wir gehen daher den aufwändigeren Weg, zu jeder Zahl von 2 bis 16 die Regeln neu aufzuschreiben, wobei sich eine offensichtliche Regelmäßigkeit ergibt.

Abkürzend schreiben wir $\langle \text{DarBa2} \rangle$ für die

$\langle \text{Darstellung zur Basis 2} \rangle$ und analog für die anderen Basen sowie $\langle \text{DarBa} \rangle$ für das Startsymbol $\langle \text{Darstellung mit Basis} \rangle$.

$\langle \text{Ziffern2} \rangle ::= 0 \mid 1$

$\langle \text{DarBa2} \rangle ::= 2 \# \langle \text{Ziffern2} \rangle \{ \langle \text{Ziffern2} \rangle \} \#$

$\langle \text{Ziffern3} \rangle ::= \langle \text{Ziffern2} \rangle \mid 2$

$\langle \text{DarBa3} \rangle ::= 3 \# \langle \text{Ziffern3} \rangle \{ \langle \text{Ziffern3} \rangle \} \#$

$\langle \text{Ziffern4} \rangle ::= \langle \text{Ziffern3} \rangle \mid 3$

$\langle \text{DarBa4} \rangle ::= 4 \# \langle \text{Ziffern4} \rangle \{ \langle \text{Ziffern4} \rangle \} \#$

... usw ...

$\langle \text{Ziffern11} \rangle ::= \langle \text{Ziffern10} \rangle \mid A$

$\langle \text{DarBa11} \rangle ::= 11 \# \langle \text{Ziffern11} \rangle \{ \langle \text{Ziffern11} \rangle \} \#$

... usw ...

$\langle \text{Ziffern16} \rangle ::= \langle \text{Ziffern15} \rangle \mid F$

$\langle \text{DarBa16} \rangle ::= 16 \# \langle \text{Ziffern16} \rangle \{ \langle \text{Ziffern16} \rangle \} \#$

$\langle \text{DarBa} \rangle ::= \langle \text{DarBa2} \rangle \mid \langle \text{DarBa3} \rangle \mid \langle \text{DarBa4} \rangle \mid \langle \text{DarBa5} \rangle \mid$

$\langle \text{DarBa6} \rangle \mid \langle \text{DarBa7} \rangle \mid \langle \text{DarBa8} \rangle \mid \langle \text{DarBa9} \rangle \mid$

$\langle \text{DarBa10} \rangle \mid \langle \text{DarBa11} \rangle \mid \langle \text{DarBa12} \rangle \mid \langle \text{DarBa13} \rangle \mid$

$\langle \text{DarBa14} \rangle \mid \langle \text{DarBa15} \rangle \mid \langle \text{DarBa16} \rangle$



Hinweis: Es gibt Varianten der EBNF.

Oft schreibt man $(...)^*$ statt $\{...\}$

und führt auch das Konstrukt $(...)^+$ ein, wenn man die 0-Iteration verbieten möchte.

Statt $::=$ wird manchmal auch einfach $=$ geschrieben.

Einzelne Terminalzeichen müssen oftmals in Anführungsstriche eingeschlossen werden.

Schlüsselwörter werden dann meist fett gedruckt.

Die Nichtterminalzeichen werden dann nicht mehr in spitze Klammern $\langle \rangle$ eingeschlossen.

Regeln werden mit einem besonderen Zeichen, meist einem Punkt, abgeschlossen.

Einschub: Überlegung 1.1.6.5: Es scheint wenig sinnvoll zu sein, fünfzehn Mal im Wesentlichen das Gleiche immer wieder hinschreiben zu müssen. Könnte man nicht ein neues Sprachelement zur EBNF hinzufügen, welches "für k von 2 bis 16" diese Wiederholungen vornimmt?!

In Ada-ähnlicher Notation *könnte* man das Sprachelement **for** $k := 2$ **to** 16 **loop** $\langle \text{Folge von Regeln} \rangle$ **end loop**

eingeführen. In den Regeln wollen wir diese Variable k benutzen, aber wir können nicht einfach " k " dort verwenden, da nicht klar ist, ob die Variable oder das Zeichen gemeint ist. Wir schließen k daher in ein Sonderzeichen (z.B. $\$$) ein und erlauben, $\$k\$$ oder allgemeiner $\$$ "Ausdruck, in dem k vorkommt" $\$$ in den Regeln zu verwenden. Dies wird am Beispiel der Darstellungen zu einer Basis klar.

Die Zahldarstellungen zur Basis $2 \leq b \leq 16$ kann man dann folgendermaßen aufschreiben, wobei wir die einzelnen Ziffern von 0 bis $b-1$ hier in der Form $(0), (1), (2), \dots, (b-1)$ notieren:

$\langle \text{Ziffern2} \rangle ::= (0) \mid (1)$

$\langle \text{DarBa2} \rangle ::= 2 \# \langle \text{Ziffern2} \rangle \{ \langle \text{Ziffern2} \rangle \} \#$

$\langle \text{DarBa} \rangle ::= \langle \text{DarBa2} \rangle$

for $k := 3$ **to** 16 **loop**

$\langle \text{Ziffern}\$k\$ \rangle ::= \langle \text{Ziffern}\$k-1\$ \rangle \mid (\$k-1\$)$

$\langle \text{DarBa}\$k\$ \rangle ::= \$k\$ \# \langle \text{Ziffern}\$k\$ \rangle \{ \langle \text{Ziffern}\$k\$ \rangle \} \#$

$\langle \text{DarBa} \rangle ::= \langle \text{DarBa} \rangle \mid \langle \text{DarBa}\$k\$ \rangle$

end loop

Die Zeile $\langle \text{DarBa} \rangle ::= \langle \text{DarBa} \rangle \mid \langle \text{DarBa}\$k\$ \rangle$ soll hier bedeuten: Es wird das alte $\langle \text{DarBa} \rangle$ um $\langle \text{DarBa}\$k\$ \rangle$ erweitert und das Ergebnis heißt wiederum $\langle \text{DarBa} \rangle$.

Hätte man das Symbol "nichts" oder "leere Menge" \emptyset zur Verfügung, so könnte man auch folgende etwas einfachere Definition verwenden:

$\langle \text{Ziffern1} \rangle ::= (0)$

$\langle \text{DarBa} \rangle ::= \emptyset$

for $k := 2$ **to** 16 **loop**

$\langle \text{Ziffern}\$k\$ \rangle ::= \langle \text{Ziffern}\$k-1\$ \rangle \mid (\$k-1\$)$

$\langle \text{DarBa}\$k\$ \rangle ::= \$k\$ \# \langle \text{Ziffern}\$k\$ \rangle \{ \langle \text{Ziffern}\$k\$ \rangle \} \#$

$\langle \text{DarBa} \rangle ::= \langle \text{DarBa} \rangle \mid \langle \text{DarBa}\$k\$ \rangle$

end loop

Problem: Was wurde nun eigentlich definiert? Nur $\langle \text{DarBa} \rangle$ oder auch $\langle \text{Ziffern3} \rangle$, $\langle \text{Ziffern4} \rangle$, $\langle \text{DarBa3} \rangle$, $\langle \text{DarBa4} \rangle$ usw.?

Warum gibt es solche Erweiterungen nicht in der Praxis?

Zunächst ist anzunehmen, dass es sehr viele solcher Erweiterungen der EBNF gibt, die von einzelnen Labors oder Firmen auf ihre Bedürfnisse zugeschnitten sind.

Das Problem ist die Bedeutung der Sprachelemente, insbesondere Eindeutigkeit und Verständlichkeit:

- Bleiben die zwischenzeitlich aufgebauten Nichtterminalzeichen erhalten?
- Wenn solche *for*-Konstrukte geschachtelt ineinander und dort in Alternativen vorkommen, ist dann das, was entsteht, stets eindeutig bestimmt?
- Und wenn man dies alles exakt definieren kann, ist der Formalismus dann noch leicht erlernbar?

Ende der Überlegung 1.1.6.5 ■

07.11.02

Kap.1.1, Informatik I, WS 02/03

227

Beispiel 1.1.6.6: Beschreibung der Ada-Anweisungen

```
<sequence_of_statements> ::= <statement> {<statement>}
<statement> ::= {<label>} <simple_statement> |
               {<label>} <compound_statement>
<simple_statement> ::= <null_statement> |
                     <assignment_statement> | <exit_statement> |
                     <procedure_call_statement> | <return_statement> |
                     <entry_call_statement> | <goto_statement> |
                     <requeue_statement> | <delay_statement> |
                     <abort_statement> | <raise_statement> | <code_statement>
```

Stopp! Unterbrechung: Man erkennt nun, warum es Varianten der EBNF gibt. In diesen Regeln treten keine Terminalzeichen auf und die spitzen Klammern sind eher unübersichtlich. Daher gehen wir nun zu der in Ada üblicheren Darstellung über (ohne spitze Klammern, Terminalzeichen werden blau und durch Anführungszeichen, Schlüsselwörter blau und durch Apostrophe hervorgehoben).

07.11.02

Kap.1.1, Informatik I, WS 02/03

228

Beispiel 1.1.6.6: Nochmals Beschreibung der Ada-Anweisungen

```
sequence_of_statements ::= statement { statement }
statement ::= { label } simple_statement |
             { label } compound_statement
simple_statement ::= null_statement |
                  assignment_statement | exit_statement |
                  goto_statement | procedure_call_statement |
                  return_statement | entry_call_statement |
                  requeue_statement | delay_statement |
                  abort_statement | raise_statement | code_statement
```

Bemerkung: Bisher haben wir hiervon nur das *null_statement* (= *skip*), das *assignment_statement* (= Wertzuweisung) und das *exit_statement* (= Verlassen einer Schleife) kennen gelernt.

07.11.02

Kap.1.1, Informatik I, WS 02/03

229

Beispiel 1.1.6.6: Ada-Anweisungen (Fortsetzung)

```
label ::= "<<" label_statement_identifier ">>"
statement_identifier ::= direct_name
null_statement ::= 'null' ";"
assignment_statement ::= variable_name "=" expression ";"
exit_statement ::= 'exit' [loop_name] ['when' condition] ";"
goto_statement ::= 'goto' label_name ";"
procedure_call_statement ::= procedure_name ";" |
                           procedure_prefix actual_parameter_part ";"
return_statement ::= 'return' [expression] ";"
```

Bemerkung: Die verbleibenden sechs *<simple_statement>* behandeln wir später bzw. im Programmierkurs.

07.11.02

Kap.1.1, Informatik I, WS 02/03

230

Beispiel 1.1.6.6: Ada-Anweisungen (Fortsetzung)

Erneuter Hinweis zur EBNF-Schreibweise:

Alle Nichtterminalzeichen, die in Normalschrift geschrieben sind, stehen irgendwo auf der linken Seite einer EBNF-Regel. Manche Nichtterminalzeichen enthalten kursiv geschriebene Teile. Diese sind bedeutungstragende Hinweise und nur das restliche Nichtterminalzeichen (ohne die kursiven Teile) tritt als linke Seite einer EBNF-Regel auf. Zwei Beispiele:

(1) *variable_name* ist ein *name*

Der Zusatz *variable* besagt, dass an dieser Stelle eine Variable stehen muss und nicht etwa ein Prozedurname oder eine Marke.

(2) *label_statement_identifier* ist ein *statement_identifier*

Der Zusatz *label* besagt, dass hier der Bezeichner für eine Marke stehen muss, die irgendwo im Programm vorhanden ist.

07.11.02

Kap.1.1, Informatik I, WS 02/03

231

Beispiel 1.1.6.6: Ada-Anweisungen (Fortsetzung)

```
compound_statement ::= if_statement | case_statement |
                      loop_statement | block_statement | accept_statement |
                      select_statement
if_statement ::= 'if' condition 'then' sequence_of_statements
               { 'elsif' condition 'then' sequence_of_statements }
               [ 'else' sequence_of_statements ] 'end if' ";"
condition ::= boolean_expression
case_statement ::= 'case' expression 'is'
                 case_statement_alternative
                 { case_statement_alternative } 'end case' ";"
case_statement_alternative ::=
    'when' discrete_choice_list "=>" sequence_of_statements
```

07.11.02

Kap.1.1, Informatik I, WS 02/03

232

Beispiel 1.1.6.6: Ada-Anweisungen (Fortsetzung)

```

loop_statement ::= [loop_statement_identifizier ":" ]
                 [iteration_scheme] 'loop'
                 sequence_of_statements
                 'end loop' [loop_identifizier] ";"
iteration_scheme ::= 'while' condition |
                  'for' loop_parameter_specification
loop_parameter_specification ::=
    defining_identifizier 'in' [reverse] discrete_subtype_definition
block_statement ::= [block_statement_identifizier ":" ]
                  ['declare' declarative_part]
                  'begin' handled_sequence_of_statements
                  'end' [block_identifizier] ";"
    
```

Dies waren 20 Regeln von etwa 400 EBNF-Regeln zur Beschreibung der Syntax von Ada 95.

Üben Sie das Lesen und Verstehen der EBNF-Regeln und damit der Syntax von Ada 95 und anderer Programmiersprachen .

Die komplette Syntax von Ada 95 finden Sie in Büchern, aber auch im Internet an mehreren Stellen, z.B. unter

<http://www.adahome.com/rm95/rm9x-P.html>

oder (einschl. Syntaxdiagrammen) unter

<http://cui.unige.ch/db-research/Enseignement/analyseinfo/Ada95/BNFindex.html>.

Die Ada-Syntax ist im "Ada Reference Manual" festgelegt; die Auflistungen der Syntax erfolgt nach dem dort benutzten Schema in 13 Abschnitten.


Ende der Anweisungen in Ada 1.1.6.6 ■

Nun fehlt nur noch die grafische Darstellung von EBNF-Regeln. Für jede linke Seite (also für jedes Nichtterminalzeichen) legen wir ein eigenes Diagramm an, das mit dem Nichtterminal bezeichnet wird.


Betrachte eine solche Regel mit dem Nichtterminal A:

$A ::= u_1 | u_2 | \dots | u_k$


Dann zeichne hierzu das Diagramm




Ist ein u_i von der Form $v_1 v_2 \dots v_r$ (Konkatenation von Zeichen), so ersetze $\rightarrow u_i \rightarrow$ durch:



Ist ein u_i oder v_j von der Form $\{ v \}$, so ersetze es durch



Ist ein u_i oder v_j von der Form $[v]$, so ersetze es durch




Zum Schluss ersetzt man bei allen Terminalzeichen und Schlüsselwörtern die rechteckigen Umrandungen durch Kreise.


1.1.6.7: Dieses Vorgehen wird hierarchisch fortgesetzt, wodurch sich schließlich für jede Grammatik, bzw. EBNF eine grafische Darstellung, das sog. **Syntaxdiagramm**, ergibt.

Um Wörter aus einem Nichtterminalzeichen A zu erzeugen, durchläuft man das zu A gehörende Syntaxdiagramm vom Eingangspfeil bis zum Ausgangspfeil. Hier gibt es i.A. viele Wege. Für jeden Weg sammelt man die Folge aller hierbei besuchten Terminalzeichen in der Durchlaufreihenfolge auf. Alle diese Zeichenfolgen bildet dann die Menge der von A erzeugten Wörter. Beispiele und eine genauere Formulierung folgen.


Beispiel: $\langle \text{Ziffer} \rangle ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9$

$\langle \text{Ziffernfolge} \rangle ::= \langle \text{Ziffer} \rangle \{ \langle \text{Ziffer} \rangle \}$





239



240

Etwas präzisere Beschreibung der Bedeutung von Syntaxdiagrammen: Gegeben sind Syntaxdiagramme und eine anfangs leere Ausgabe.

Ziel ist es, das Diagramm, das zum Startsymbol gehört, in Richtung der Pfeile vom Eingangsfeil bis zum Ausgangspfeil zu durchlaufen. Trifft man hierbei auf Zeichen in Kreisen, so schreibt man diese in der Durchlaufreihenfolge hinter die bereits vorhandene Ausgabe. Trifft man auf ein Rechteck mit dem Nichtterminalzeichen A, so klebt man das zu A gehörende Diagramm an dieser Stelle ein und durchläuft das neu entstandene Diagramm weiter. (Gibt es kein zu A gehörendes Diagramm, so bricht man ergebnislos ab.)

Alle möglichen Ausgaben, die zu einem vollständigen Durchlauf vom Eingangsfeil bis zum Ausgangspfeil des Startsymbol-Diagramms gehören, bilden die erzeugte Sprache.


07.11.02

Kap.1.1, Informatik I, WS 02/03

241

Beispiel

if_statement ::= 'if' condition 'then' sequence_of_statements
 {'elsif' condition 'then' sequence_of_statements}
 ['else' sequence_of_statements] 'end if' ";"



07.11.02

Kap.1.1, Informatik I, WS 02/03

242